

LOST FORREST

This program will fit into an unexpanded VZ-200 but only just and spaces should be missed out where they are not necessary.

The Player can move from one location to another by using the commands:

N - NORTH
S - SOUTH
W - WEST
E - EAST

The command 'HELP' will display a list of verbs (or words) that the computer understands.

These words will aid the player in his search for the missing 'Green God' and several other treasures.

When the player thinks he has all of the missing treasures he should return to the location of the 'large trees' and type the command 'SCORE'. If all the treasures have been found then a winning message will be displayed.

```

60 CLEAR100
70 V=17:W=18:G=14
80 GOSUB1300
90 CLS:PRINT"LOST FOREST"
100 PRINT"-----"
105 PRINT"MOVE NO. ";M:IFM>50THEN4000
110 PRINT"YOUR LOCATION"
120 PRINTD$(RM)
130 PRINT"EXITS: ";
140 FORI=1TOLEN(R$(RM)):PRINTMID$(R$(RM),I,1);", ";:NEXTI
170 PRINT:FORI=1TOG
190 IFL(I)=RMANDF(I)=0THENPRINT"YOU CAN SEE ";O$(I);" HERE"
200 NEXTI
210 PRINT"===== ":PRINTM$:M$="WHAT"
230 INPUT"WHAT WILL YOU DO NOW";Q$
240 V$="":W$="":VB=0:OB=0
250 FORI=1TOLEN(Q$)
260 IFMID$(Q$,I,1)=" "ANDV$=""THENV$=LEFT$(Q$,I-1)
270 IFMID$(Q$,I+1,1)<>" "ANDV$<>" "THENW$=MID$(Q$,I+1,LEN(Q$)-1)
275 IFMID$(Q$,I+1,1)<>" "ANDV$<>" "THENI=LEN(Q$)
280 NEXTI
290 IFW$=""THENV$=Q$
300 FORI=1TOV:IFV$=V$(I)THENVB=I
310 NEXTI
330 FORI=1TOW:IFW$=O$(I)THENOB=I
340 NEXTI
360 IFW$=""ANDOB=0THENM$="THAT'S SILLY"
370 IFVB=0THENVB=V+1
380 IFW$=""THENM$="I NEED TWO WORDS"
390 IFVB>VANDOB>0THENM$="YOU CAN'T '"+Q$+"'"
400 IFVB>VANDOB=0THENM$="YOU DON'T MAKE SENSE"
410 IFVB<VANDOB=0ANDC(OB)=0THENM$="YOU DON'T HAVE '"+W$+"'"
460 IFVB=1THENGOSUB500ELSEIFVB=0THENGOSUB570
462 IFVB=3ORVB=4ORVB=5ORVB=6ORVB=7THENGOSUB640
466 IFVB=8ORVB=9THENGOSUB980
470 IFVB=10THENGOSUB1030ELSEIFVB=11THENGOSUB1060
472 IFVB=12THENGOSUB1100ELSEIFVB=13THENGOSUB1120
474 IFVB=14THENGOSUB1150ELSEIFVB=15THENGOSUB1170
476 IFVB=16THENGOSUB1210ELSEIFVB=17THENGOSUB1230
478 IFVB=18THENGOSUB1290
490 GOTO90
500 PRINT"WORDS I KNOW: "
510 FORI=1TOV:PRINTV$(I);", ";:NEXTI
540 M$="":PRINT:GOSUB1280
560 RETURN
570 PRINT"YOU ARE CARRYING: "
580 FORI=1TOG:IFC(I)=1THENPRINTO$(I);", ";
590 NEXTI
610 M$="":PRINT:GOSUB1280
630 RETURN
640 D=0:IFOB=0THEND=VB-3
780 IFRM=6ANDF(8)<>2THENPRINT"YOU FALL ":GOTO3500
785 IFRM=6ANDF(8)=2ANDD=4THENF(8)=0
790 IFF(12)=0ANDRM=8THENPRINT"THE EAGLE ATTACKS":GOTO3000
800 IFRM=1ANDF(11)=0THENPRINT"HAWK ATTACKS SAVAGELY":GOTO3000

```

COMPUTER INPUT

Feb 85 p 27.

1 of 2.


```

850 RL=LEN(R$(RM)):F(17)=0
860 FORI=1TORL:U$=MID$(R$(RM),I,1)
880 IF(U$="N"ANDD=1ANDF(17)=0)THENRM=RM-3:M=M+1:F(17)=1
890 IF(U$="S"ANDD=2ANDF(17)=0)THENRM=RM+3:M=M+1:F(17)=1
900 IF(U$="W"ANDD=3ANDF(17)=0)THENRM=RM-1:M=M+1:F(17)=1
910 IF(U$="E"ANDD=4ANDF(17)=0)THENRM=RM+1:M=M+1:F(17)=1
920 NEXTI
930 M$="OK"
935 IFF(17)=0THENM$="YOU CAN'T GO THAT WAY"
940 IFD<1THENM$="GO WHERE"
970 RETURN
980 IFOB>9THENM$="I CAN'T GET "+W$:RETURN
982 IF(OB=1OROB=2)ANDC(6)=0THENM$="YOU NEED A CLOTH BAG":RETURN
985 IFL(OB)<>RMTHENM$="IT'S NOT HERE"
990 IFF(OB)<>0THENM$="WHAT "+W$+"?"
1000 IFC(OB)=1THENM$="YOU ALREADY HAVE IT"
1010 IFOB>0ANDL(OB)=RMANDF(OB)=0THENC(OB)=1:M$="YOU HAVE "+W$
1011 IFOB>0ANDL(OB)=RMANDF(OB)=0THENF(OB)=1
1020 RETURN
1030 IFRM=8ANDOB=5ANDC(5)=1ANDF(12)=0THENM$="IT FLYS OFF!!"
1031 IFRM=8ANDOB=5ANDC(5)=1ANDF(12)=0THENF(12)=1:C(5)=0
1040 IFRM=1ANDOB=5ANDC(5)=1ANDF(11)=0THENM$="MISSED!!!":C(5)=0
1046 IFOB=7ANDC(7)=1THENM$="IT BREAKS, OBJECT FALLS OUT"
1047 IFOB=7ANDC(7)=1THENF(1)=0:C(7)=0:L(1)=RM
1048 IFC(OB)=1THENM$="WHAT A TEMPER!!!":C(OB)=0
1050 RETURN
1060 IFRM=1ANDOB=11ANDC(4)=1THENM$="IT FLYS OFF WITH BERRIES"
1070 IFRM=1ANDOB=11ANDC(4)=1THENF(11)=1:C(4)=0:RETURN
1090 RETURN
1100 IFRM=3ANDOB=10ANDF(10)=0THENM$="YOU MAY NEED THIS"
1101 IFRM=3ANDOB=10ANDF(10)=0THENF(6)=0:F(10)=1
1110 RETURN
1120 IFOB=13ANDRM=9THENM$="OK":RM=6:M=M+1
1140 RETURN
1150 IFRM=6ANDOB=8ANDC(8)=1THENM$="TRY THIS EXIT":F(8)=2:C(8)=0
1160 RETURN
1170 IFRM=17ANDOB=14ANDC(9)=1ANDF(14)=0THENM$="YOU DIG A HOLE"
1175 IFRM=17ANDOB=14ANDC(9)=1ANDF(14)=0THENF(14)=1:F(3)=0
1180 RETURN
1210 IFC(OB)=1THENC(OB)=0:L(OB)=RM:F(OB)=0:M$="DONE"
1220 RETURN
1230 S=0:FORI=1TOG:IFC(I)=1THENS=S+10
1240 NEXTI
1271 IFS=>50ANDC(1)*C(2)*C(3)*C(6)*C(9)<>0THENPRINT"WELL DONE ";
1272 IFS=>50ANDC(1)*C(2)*C(3)*C(6)*C(9)<>0ANDRM=16THEN1274
1273 PRINT"SCORE SO FAR=";S:GOTO1280
1274 SC=(50-M)+S+1:PRINT"YOUR SCORE=";SC
1275 PRINT"YOU'VE WON!!!":GOTO3020
1280 INPUT"PRESS RETURN TO CONTINUE";CT$
1290 RETURN
1300 DIMR$(17),D$(17),O$(W),V$(V),C(W),L(G),F(W)
1320 DATA13,0,17,15,11,3,13,6,6,3,1,8,9,17
1330 FORI=1TOG:READL(I):NEXTI
1360 DATAHELP,INV,G0,N,S,W,E,GET,TAKE,THROW,FEED,ROLL,CLIMB
1365 DATADROP,DIG,LEAVE,SCORE
1380 FORI=1TOV:READV$(I):NEXTI
1410 DATAE,WE,SW,E,SWE,NSW,E,NE,NSW,SE,SWE,NW,NE,NSWE,SW,E
1420 DATANW,N
1430 FORI=0TO17:READR$(I):NEXTI
1520 DATATREES,MORE TREES,EVEN MORE TREES,ANOTHER TREE,TREES
1530 DATAA TREE,UP TREE,TREE,TALL TREES,GREEN TREE,MORE TREES
1540 DATAOAK TREE,BIG TREE,OVERGROWN TREE,SHORT TREE
1550 DATATREES,LARGE TREES,MANY TREES
1580 FORI=0TO17:READD$(I):NEXTI
1610 DATAGREEN GOD,LAURELS,GOLD LEAF,BERRIES,ACORN,SACK
1620 DATABEEHIVE,ROPE,TROWEL,LOG,HAWK,EAGLE,GREEN TREE
1630 DATATREE,NORTH,SOUTH,EAST,WEST
1640 FORI=1TOW:READO$(I):NEXTI
1670 F(1)=1:F(3)=1:F(6)=1:RM=RND(18)-1
1675 IFRM<9THEN1670
1680 M=1:RETURN
3000 FORI=1TO2000:NEXTI:CLS:PRINT"YOU ARE NOW OFFICIALLY DEAD"
3010 PRINT"YOU HAVE NOT MANAGED TO COMPLETEYOUR FOREST WALK"
3020 PRINT:INPUT"DO YOU WANT TO PLAY AGAIN";PA$
3030 ILEFT$(PA$,1)="Y"THEN60
3040 CLS:END
3500 FORI=1TO2000:NEXTI:CLS
3510 PRINT"YOU'VE LANDED IN A WOLF TRAP"
3520 PRINT:GOTO3010
4000 CLS:PRINT"YOUR TIME HAS RUN OUT AND'YOU ARE TOO WEAK TO";
4010 PRINT"CONTINUE":GOTO3010

```